

Basic program gives high resolution plus X and Y-axes

Curve-Plotting with your Sorcerer

This function plotting program for the Exidy Sorcerer takes advantage of the computer's user-definable graphics feature to produce a high resolution display of any single-variable function. The function and the domain over which it will be plotted can be defined by the user, and the final display is most impressive.

by DANIEL WONG


The program will plot any single variable function with a high degree of resolution, and can be run on any Sorcerer computer. It produces a display which is almost continuous, without the "stair-case" effect seen in lower resolution plotting programs. The X and Y-axes and their corresponding minimum and maximum values are also depicted.

To run the program the user only has to define the function to be plotted in statement 10 and the domain of the X-axis in statement 20. XI is the minimum value that X can take, and XA is the maximum value. As listed here the program will plot the function $X=\text{SIN}(X)$ between the limits $X=0$ and $X=6.3$ (in radians in this case).

Note line 1225, which confines the

possible positions of the plotted points between the values of -3968 and -2049, which are the limits of the screen RAM of the Sorcerer (in decimal). Poking data into memory locations outside these limits can have strange results, and may cause the programs to crash. Adding this simple test to any program that manipulates the screen RAM can save many hours of de-bugging.

Any single variable function can be plotted. $X=\text{TAN}(X)$ gives interesting results, as does $X=\text{EXP}(X)$. Remember that when you define a new function to be plotted it will also be necessary to change the limits of the X-axis. The program will automatically scale the plot to make best use of the available screen area.



Above is a typical plot from the program listed on the opposite page.

Users of the Sorcerer will be aware of the potential of the machine's user definable graphics and memory mapped display. Theoretically the combination of these two features results in a display with a resolution of 512 x 240 dots. The catch is that the programming for high resolution is quite complex. The subroutine from line 1020 to line 1245 in this program shows what must be done. Study of this routine will provide valuable hints to anyone interested in fully exploiting the Sorcerer's high resolution capabilities.

```

10 DEF FNY(X)=SIN(X)
20 XI=0:XA=6.3
90 DIM BK(24),X1(7),Y1(7)
100 GOSUB 900
101 PRINT CHR$(12)
102 IF YI=YA AND YI<0 THEN 128
104 IF YI=YA AND YI>0 THEN 126
110 IF YI=0 OR (YI>0 AND YA>0) THEN 140
111 IF YA=0 OR (YI<0 AND YA<0) THEN 131
120 XX=INT(ABS(YI)*25/(YA-YI))
124 GOTO 150
126 YI=0:YA=2*YA:GOTO 140
128 YA=0:YI=2*YI
131 XX=24:GOTO 150
140 XX=-1
150 FOR I=0 TO 50
160 H=I-64*XX-2227
170 POKE H,45
180 NEXT I
230 IF XI=0 OR (XI>0 AND XA>0) THEN 260
231 IF XA=0 OR (XI<0 AND XA<0) THEN 251
240 YX=INT(ABS(XI)*51/(XA-XI))
250 GOTO 270
251 YX=50:GOTO 270
260 YX=-1
270 FOR J=0 TO 24
280 H=YX-64*J-2227
290 POKE H,124
295 NEXT J
300 GOSUB 1020
301 AA=XI:BB=1:XX=XX-1:GOSUB 350
302 AA=INT(2.5*(XA-XI)+0.5)/10+XI:BB=13:GOSUB 350
303 AA=INT(5*(XA-XI)+0.5)/10+XI:BB=26:GOSUB 350
304 AA=INT(7.5*(XA-XI)+0.5)/10+XI:BB=39:GOSUB 350
305 AA=XA:BB=51:GOSUB 350
310 AA=YI:BB=YX:XX=0:GOSUB 350
321 AA=INT(2.5*(YA-YI)+0.5)/10+YI:XX=6:GOSUB 350
322 AA=INT(5*(YA-YI)+0.5)/10+YI:XX=12:GOSUB 350
323 AA=INT(7.5*(YA-YI)+0.5)/10+YI:XX=18:GOSUB 350
330 AA=YA:XX=24:GOSUB 350
335 END
350 CC$=STR$(AA)
360 XL=LEN(CC$)
370 FOR I=1 TO XL
380 DD=ASC(RIGHT$(CC$,I))
390 H=BB-I-64*XX-2227
400 POKE H,DD
410 NEXT I
420 RETURN
900 PRINT CHR$(12)
905 PRINT "Calculation in progress. Please wait"
910 DIM Y2(50)
915 FOR J=0 TO 50
920 X=J*(XA-XI)/50
925 Y2(J)=FNY(X+XI)
930 IF J=0 THEN 985
935 K=J-1
940 FOR I=0 TO K
945 IF Y2(J)>=Y2(I) THEN 980
950 C=Y2(I):D=Y2(J)
955 FOR L=I TO K
960 A=Y2(L+1):Y2(L+1)=C:C=A
965 NEXT L
970 Y2(I)=D
975 GOTO 985
980 NEXT I
985 NEXT J
990 YI=INT(Y2(0))
995 YA=INT(Y2(50)+0.5)
996 RETURN
1020 P=1024:Q=128
1030 FOR K=-1024 TO -1:POKE K,0:NEXT K
1040 FOR I=0 TO 50
1050 A=I*8:B=I*8+7
1060 FOR J=A TO B
1070 C=J-A
1080 X=J*(XA-XI)/408
1090 YY=FNY(X+XI)-YI
1100 Y=INT(YY*200/(YA-YI)+0.5)
1110 BK(C)=INT(Y/8)
1120 IF C=0 THEN 1160
1130 D=C-1
1140 IF BK(C)=BK(D) THEN 1160
1150 P=P-8:Q=Q+1
1160 X1(C)=J-8*INT(J/8)
1170 Y1(C)=Y-8*INT(Y/8)
1180 E=2^(7-X1(C))
1190 F=7-P-Y1(C)
1191 IF F>=0 THEN 1245
1200 G=E+PEEK(F)
1210 POKE F,G
1220 H=I-64*BK(C)-2227
1225 IF H<-3968 OR H>-2049 THEN 1235
1230 POKE H,Q
1235 NEXT J
1240 P=P-8:Q=Q+1
1242 NEXT I
1245 RETURN

```